MAT4161/MAT4181 - Cálculo a uma Variável - 2023.1

P2 Maple - Versão II - Gabarito

Questão 1

Seja f a função definida por $f(x) = \sin(\ln(x^2 + 1))$.

Considere como domínio de f o **maior intervalo possível** para que f seja inversível e para que o gráfico de f^{-1} (a função inversa de f) passe pelo ponto $P = (f(-1), -1)$.

(a) Determine o domínio de f.

```
> restart;
> f:=x->sin(ln(x^2+1));
```

$$f := x \rightarrow \sin(\ln(x^2 + 1)) \qquad (1)$$

```
> plot(f(x),x=-3..1);
```




```
> s:=solve(D(f)(x)=0);
```

$$s := 0, \sqrt{e^{\frac{1}{2}\pi} - 1}, -\sqrt{e^{\frac{1}{2}\pi} - 1} \qquad \textbf{(2)}$$

```
> evalf(s[3]);
```

$$-1.952044411 \qquad \textbf{(3)}$$

Como $P = (f(-1), -1)$ deve pertencer ao gráfico de f^{-1}, então $Q = (-1, f(-1))$ deve pertencer ao gráfico de f. Ou seja, $x = -1$ deve pertencer ao domínio de f.

O maior intervalo contendo $x = -1$ onde a derivada de f não muda de sinal é

$$Dom(f) = \left[-\sqrt{e^{\frac{\pi}{2}} - 1}, 0\right] = [-1.952, 0].$$

(b) Determine uma expressão para f^{-1}.

```
> t:=solve(x=f(y),y);
```

$$t := \sqrt{-1 + e^{\arcsin(x)}}, -\sqrt{-1 + e^{\arcsin(x)}} \qquad \textbf{(4)}$$

Como determinar a expressão correta? Lembre que o ponto $P = (f(-1), -1)$ deve pertencer ao gráfico de f^{-1}:

```
> simplify(subs(x=f(-1),t[1]));
```

$$1 \tag{5}$$

```
> simplify(subs(x=f(-1),t[2]));
```

$$-1 \tag{6}$$

Logo, a expressão correta é a segunda: $f^{-1}(x) = -\sqrt{-1 + e^{\arcsin(x)}}$.

```
> finv:=unapply(t[2],x);
```

$$finv := x \to -\sqrt{-1 + e^{\arcsin(x)}} \tag{7}$$

(c) Considere a região plana $\mathcal{R}$ delimitada pela reta $y = -\frac{3}{4} - x$ e pelos gráficos de f e f^{-1}.

(c.1) Esboce a **região** $\mathcal{R}$.

```
> with(plots):
> Graff:=plot(f(x),x=-1.952..0,y=f(0)..f(-1.952),color=red):
> Graffinv:=plot(finv(x),x=f(0)..f(-1.952),y=-1.952..0,color=blue):
> Retas:=plot([-3/4-x,x],x=-1.952..f(-1.952),color=[magenta,black])
  :
> display(Graff,Graffinv,Retas);
```



Esboço da **região**:



(c.2) Calcule a área da região $\mathcal{R}$ utilizando integrais.

```
> a:=fsolve(f(x)=-3/4-x);
```

$$a := -1.733211645 \tag{8}$$

```
> b:=fsolve(finv(x)=-3/4-x);
```

$$b := 0.9832116449 \tag{9}$$

```
> Area:=Int(f(x)-(-3/4-x),x=a..0)+Int(finv(x)-(-3/4-x),x=0..b);
```

$$Area := \int_{-1.733211645}^{0} \left(\sin\left(\ln\left(x^2+1\right)\right) + \frac{3}{4} + x\right) \mathrm{d}x + \int_{0}^{0.9832116449} \left(-\sqrt{-1+\mathrm{e}^{\arcsin(x)}} + \frac{3}{4} + x\right) \mathrm{d}x \tag{10}$$

```
> value(Area);
```

$$1.068770821 \tag{11}$$

Questão 2

```
> restart;
> f1:=x->ln((2*x^2-6*x+4)/(x+2)^2);
```

$$f1 := x \mapsto \ln\left(\frac{2\,x^2-6\,x+4}{(x+2)^2}\right) \tag{12}$$

```
> f2:=x->(sqrt(x^2+ln(x)+2)-x)/(x-1);
```

$$f2 := x \mapsto \frac{\sqrt{x^2+\ln(x)+2}-x}{x-1} \tag{13}$$

```
> with(plots):
> A:=plot(f1(x),x=-10..0,color=blue):
> B:=plot(f2(x),x=0..10,color=red):
> display(A,B);
```



Item a) Domínio da função

Para a função $f1$:

```
> solve(2*x^2-6*x+4>0);
```

$$(-\infty, 1), (2, \infty) \tag{14}$$

```
> solve((x+2)^2=0);
```

$$-2, -2 \tag{15}$$

Como seu domínio é da forma $(-\infty, a) \cup (a, 0]$, segue que $a = -2$.

Para a função $f2$:

```
> solve(x>0);
```

$$(0, \infty) \tag{16}$$

```
> solve(x^2+ln(x)+2>=0);
```

$$\left(e^{-\frac{\mathrm{LambertW}\left(\frac{2}{e^4}\right)}{2} - 2}, \infty\right) \tag{17}$$

```
> b:=exp(-(1/2)*LambertW(2/exp(4))-2);
```

$$b := e^{-\frac{\mathrm{LambertW}\left(\frac{2}{e^4}\right)}{2} - 2} \tag{18}$$

```
> evalf(b);
```

$$0.1329636711 \tag{19}$$

```
> solve(x-1=0);
```

$$1 \tag{20}$$

Como seu domínio é da forma $(\mathrm{b}, c) \cup (c, +\infty]$, segue que $b = 0.132$ e $c = 1$.

Item b) Assíntotas horizontais:

```
> limit(f1(x),x=-infinity);
```

$$\ln(2) \tag{21}$$

```
> limit(f2(x),x=infinity);
```

$$0 \tag{22}$$

Logo, as assíntotas horizontais são $y = \ln(2)$ e $y = 0$.

Item c) Assíntotas verticais:

```
> limit(f1(x),x=-2);
```

$$\infty \tag{23}$$

```
> L:=limit(f2(x),x=b,right);
```

$$L := \frac{\dfrac{\sqrt{4\left(e^{-\frac{\mathrm{LambertW}\left(\frac{2}{e^4}\right)}{2} - 2}\right)^2 - 2\,\mathrm{LambertW}\left(\frac{2}{e^4}\right)}}{2} - e^{-\frac{\mathrm{LambertW}\left(\frac{2}{e^4}\right)}{2} - 2}}{e^{-\frac{\mathrm{LambertW}\left(\frac{2}{e^4}\right)}{2} - 2} - 1} \tag{24}$$

```
> evalf(L);
```

$$0.1533542098 \tag{25}$$

```
> limit(f2(x),x=1,left);
```

$$-\infty \tag{26}$$

Logo, as assíntotas verticais são $x = -2$ e $x = 1$.